

RELOGO GETTING STARTED GUIDE

JONATHAN OZIK - REPAST DEVELOPMENT TEAM

0. BEFORE WE GET STARTED

Before we can do anything with ReLogo, we need to make sure that we have a proper installation of Repast Symphony 2.0. Instructions on downloading and installing Repast Symphony on various platforms can be found on the Repast website.

1. GETTING STARTED WITH RELOGO

Now let us begin our exploration of ReLogo. We will be building a simple agent-based model involving zombies chasing humans and humans running away from zombies. Our approach will be to not overwhelm you but to explain only as much as is needed at each step. By the end of this chapter, we'll have covered a lot of ground and you'll be able to continue with your own explorations of ReLogo.

The first thing we must do is create a new ReLogo project. This is done by clicking on the New ReLogo Project icon in the toolbar (Fig. 1) at the top of our ReLogo workspace¹.




FIGURE 1. The New ReLogo Project icon.

This brings up the New ReLogo Project Wizard (Fig. 2) which gives us the ability to name our project (and a few more options which we'll ignore for now). Typing in "Zombies" in the "Project name" field², we press the "Finish" button and the wizard sets

Date: January 18, 2012.

¹The default workspace that came with your Repast Symphony installation should be set up correctly. However, if you have modified the settings or are using a custom workspace see Section 2.7 for the preferred ReLogo workspace settings.

²ReLogo differentiates between capitalized and uncapitalized letters (i.e., it is case sensitive) so, when following this tutorial, it will be important to note the capitalization of names, variables, etc.

up our project. The project structure should look like Fig. 3 (but if it does not, see this note³).




FIGURE 2. The New ReLogo Project Wizard.

What we see is the Zombies project folder, the “src” subfolder, the “zombies.relogo” package, and the relevant ReLogo files, in addition to a “shapes” folder. More on each of these things as we proceed.

1.1. Creating the Human and Zombie turtle types. Now that we have all our model infrastructure in place, we can start specifying the details of our Zombie model. First, since zombies love to chase humans, we create the Human turtle⁴ type⁵. We do this by selecting the “zombies.relogo” package, if it isn’t selected, and then clicking on the New Turtle icon (Fig. 4) in the toolbar⁶. This brings up the New Turtle Wizard (Fig. 5) which allows us to specify the name of our turtle type (Human). If we initially selected the “zombies.relogo”

³Here we assume that your ReLogo Resource Filter is enabled. If the ReLogo Resource Filter is disabled, you may see more elements in your Zombies project. See Section 2.4 on how to disable/enable this filter.

⁴In Logo dialects a “turtle” is a mobile agent.

⁵Many of the ReLogo entities we’ll encounter, including turtle types, are what are known in Object Oriented programming languages as *classes*.

⁶Selecting the “zombies.relogo” package simplifies the next step.




FIGURE 3. The directory structure of the newly created Zombies project.




FIGURE 4. The New Turtle icon.

package, we simply fill in the Name field with “Human” and hit the Finish button⁷. At this point we should be greeted by our newly created Human turtle type as seen in Fig. 6⁸

⁷If the “zombies.relogo” package hadn’t been selected, we need to fill in the Package field with “zombies.relogo” before hitting the Finish button.

⁸For those curious about the .groovy file ending, ReLogo is an agent-based modeling domain specific language (ABM DSL), written in the Groovy programming language. While it isn’t necessary to be able to follow this getting started guide, we recommend getting to know the language through the Groovy website and the many Groovy books which are available. There is also a very convenient web based console where you can experiment with Groovy.




FIGURE 5. The New Turtle Wizard with information for creating the Human turtle type.




FIGURE 6. The view after creation of the Human turtle type.




FIGURE 7. The New Turtle Wizard with information for creating the Zombie turtle type.

Next we follow a similar procedure to create the Zombie turtle type (Fig. 7).

1.2. Defining Human and Zombie behaviors. The next step in building our model is defining the behaviors of the Human and Zombie turtle types. We'll dive right in.

Each Human will have a *step* method which looks like this (but first, see important note⁹):

```

1  def step(){
2      def winner = minOneOf(neighbors()){
3          count(zombiesOn(it))
4      }
5      face(winner)
6      forward(1.5)
7      if (infected){
8          infectionTime++
9          if (infectionTime >= 5){
10             hatchZombies(1){
11                 size = 2
12             }
13             die()
14         }
15     }
16 }
```

LISTING 1. Human step method.

The thinking here is that at every time advancement of our simulation, or “tick,” each Human will execute this method. The Human first chooses a *winner* which is, in plain English, one of the neighboring patches¹⁰ with the fewest number of Zombies on it. When the winner is chosen, the Human *faces* it and moves *forward* 1.5 steps, thereby running away from potential high Zombie areas. If the Human is *infected*, and 5 or more time ticks have passed since the initial infection, the Human *dies* and *hatches* a Zombie.

Let's briefly review the code. A ReLogo turtle has a number of ReLogo primitives¹¹, or capabilities, it can use, without having to create its own. Among these are *minOneOf* and

⁹For those following this getting started guide electronically, simple copying-and-pasting of the code from the tutorial will result in errors. Even if the line numbers are removed there can be errors resulting from the formatting of quoted strings and other elements. In short, to get the most out of the guide with the least amount of errors, we recommend typing in the code yourself.

¹⁰Patches were introduced by StarLogo and are square shaped immobile agents which make up an underlying grid structure in a ReLogo world.

¹¹The set of available primitives can be found in the ReLogoPrimitives.html file that came with the Repast Symphony distribution. The primitives are separated broadly by the type of ReLogo entity that uses them and further by the type of primitive category (e.g, motion, rotation, etc.). Clicking on a primitive will give you more details. This will likely be your primary reference as you explore ReLogo so it will help to get familiar with it.

neighbors. *minOneOf* takes as an argument a set of “things” and a block defining what quantity will be used to determine the minimum of.

The “things” (in line 2 of Listing 1) are the set of patches returned by the *neighbors* primitive, which are the 8 neighbors¹² of the patch that the Human is on. You’ll notice that *neighbors* is followed by a set of empty parentheses. This is because the primitives are what are referred to in some programming circles as *methods* and the parentheses hold the arguments to those *methods*. If the method doesn’t take any arguments, as is the case with the *neighbors* primitive, we still need to call the method but without any arguments. As mentioned above, *minOneOf* takes two arguments, the second being a block of code, specified between curly braces. Groovy allows, for clarity, to omit the parentheses around a block of code if it’s the last argument to a method. So instead of writing:

```
maxOneOf(neighbors() , {
    count(zombiesOn(it))
})
```

we can write:

```
maxOneOf(neighbors()){
    count(zombiesOn(it))
}
```

When you define a turtle type in ReLogo, there are primitives that are automatically made available to various ReLogo entities¹³. *zombiesOn* is such a primitive that was made available to turtles when we defined the Zombie turtle type. It takes as an argument a patch¹⁴ and returns the Zombie turtles on that patch. The *it* is another feature of Groovy. It is an implicit argument to the block of code¹⁵ which allows us to simplify:

¹²This is referred to as the Moore neighborhood in a grid.

¹³For a complete list, see Table 1 in Appendix A.

¹⁴*zombiesOn* can also take a turtle as an argument, which has the same semantics as the version that takes a patch as an argument, but in the former case the patch is the patch under the turtle.

¹⁵Arguments passed to a block of code are a comma separated sequence of variable names followed by `->`. For any block of code without input arguments explicitly specified, the assumption is that it takes one implicit parameter named *it*.

```

maxOneOf(neighbors()){ p ->
    count(zombiesOn(p))
}

```

to:

```

maxOneOf(neighbors()){
    count(zombiesOn(it))
}

```

The primitive *count* can be applied to any set of items and returns the number of items in the set. Finally, the *def* keyword in Groovy is used when we define variables or the return types of methods. It's basically a wildcard indicating that whatever it adorns is of "some" type, without specifying further¹⁶. Thus, taken together, lines 2-4 of Listing 1, *assign* to the variable *winner* the patch with the fewest number of Zombies on it¹⁷.

We'll assume that lines 5 and 6 of Listing 1 are self explanatory, and proceed to the *conditional* statement on lines 7-13. The *if* keyword is commonly used in many programming languages to determine the logical flow of some statements. In this case, we are checking to see if the Human is *infected* and if so, we proceed to line 8 and otherwise we skip down past line 13, the end of the *if* block. This is a good time to introduce the fact that in addition to methods, ReLogo entities have properties as well. Thus, elsewhere (which we'll show in Listing 2), we've explicitly specified that the Human turtle type has an *infected* property which we will initially set to *false*. The same holds for the *infectionTime* property¹⁸, on line 8, which we'll initially set to 0, again elsewhere (Listing 2).

On lines 9-12, we check to see if the *infectionTime* is greater than or equal to 5 and, if so, the Human *hatches* a Zombie and then it *dies*¹⁹.

Now let's see what our full Human turtle type looks like, with the *step* method and the turtle properties:

¹⁶Groovy, for those interested, uses dynamic (but strict) typing. This is just one of the reasons why we often refer to it as a "less neurotic" Java.

¹⁷One of the most common mistakes is replacing the equality operator `==` with the assignment operator `=`. In the former case, the *equality* of the two sides is checked and either a *true* or *false* is returned. In the latter case, the right hand side is *assigned* to the left hand side.

¹⁸The `infectionTime++` notation is a shortcut for `infectionTime = infectionTime + 1`.

¹⁹The order might be confusing but if the Human dies and is removed from the simulation before hatching the Zombie, well, it can't hatch the Zombie!


```
1  // package declaration and imports, which we can ignore for now
2
3  class Human extends BaseTurtle {
4
5      def infected = false
6      def infectionTime = 0
7
8      def step(){
9          def winner = minOneOf(neighbors()){
10              count(zombiesOn(it))
11          }
12          face(winner)
13          forward(1.5)
14
15          if (infected){
16              infectionTime++
17              if (infectionTime >= 5){
18                  hatchZombies(1){
19                      size = 2
20                  }
21                  die()
22              }
23          }
24      }
25
26  }
```

LISTING 2. The Human turtle type.

The takeaway here is that turtle type properties and methods are defined within the *class* body of the turtle type, in this case between the curly braces on lines 3 and 26.

Now let's move on to the Zombie turtle type. It looks like this:

```

1  // package declaration and imports, which we can ignore for now
2
3  class Zombie extends BaseTurtle {
4
5      def step(){
6          def winner = maxOneOf(neighbors()){
7              count(humansOn(it))
8          }
9
10         face(winner)
11         forward(0.5)
12
13         if (count(humansHere()) > 0){
14             label = "Brains!"
15             infect(oneOf(humansHere()))
16         }
17         else {
18             label = ""
19         }
20     }
21
22     def infect(human){
23         human.infected = true
24     }
25 }

```

LISTING 3. The Zombie turtle type.

What should be noticed here is that, in addition to the method *step*, we’ve defined another auxiliary method *infect* which takes one argument. Let us take a moment to explore how it’s used. Looking at lines 13-15 in Listing 3, we see that we check using the primitives *count* and *humansHere*²⁰ if there are any Human turtle types “here.” If there are, we define the default turtle type property *label* to be “Brains!” and proceed to *infect oneOf* the *humansHere*²¹. To infect the human, we access the human’s property *infected* by *referencing* it with a period²².

²⁰As you probably guessed, this is one of the generated primitives when we defined the Human turtle type.

²¹As you likely notice, the purpose of Logo constructs in general and ReLogo code in particular can often easily be understood and can lead to more manageable code.

²²In Object Oriented languages, accessing an object’s properties and methods is a common idiom.




FIGURE 8. The UserObserver.groovy file in the Package Explorer view.

1.3. Coordinating behaviors with the UserObserver. At this point we have both the Human and Zombie turtle types specified. What we need next is to define the overall flow of our model. Logo uses the notion of an “observer” to accomplish this role. The UserObserver is the default observer type that was made for us when we created the Zombies project. Double clicking the UserObserver.groovy file in the Package Explorer view (Fig. 8) will bring up the file.

A common idiom is to define a method that sets up our simulation. We do so by defining the *setup* method:

```

1  def setup(){
2      clearAll()
3      setDefaultShape(Human, "person")
4      createHumans(numHumans){
5          setxy(randomXcor(),randomYcor())
6      }
7      setDefaultShape(Zombie, "zombie")
8      createZombies(numZombies){
9          setxy(randomXcor(),randomYcor())
10         size = 2
11     }
12 }

```

LISTING 4. The UserObserver setup method.

We'll first describe, in plain English, what this piece of code does. It is a good idea in any initialization code to reset the simulation, so we first *clearAll* existing entities from the world. Then we define the default shapes of both the Humans and Zombies and create a different number of each (with primitives generated by defining the Human and Zombie turtle types). We also scatter the created turtles randomly across our world and set the size of the Zombies to 2, to make them more prominent.

Now that we know the gist of Listing 4, let's delve a little deeper. The *clearAll* primitive removes all existing entities from the ReLogo world and resets the patches to their default state. The *setDefaultShape* primitive takes two arguments. The first argument is a turtle type and the second is a *string*²³ specifying a turtle shape. To understand what shapes are available we can open the "shapes" folder in the Package Explorer view (Fig. 9) and see the default shapes that come with any newly created ReLogo project²⁴. Simply specifying the name of one of these shapes (without the .svg suffix) in the *setDefaultShape* primitive's second argument will set the chosen turtle types default shape to it²⁵.

Another thing that the astute reader might notice is that there are references to *numHumans* and *numZombies*. While it's possible for these to be properties of the UserObserver class, we will opt to make these "tweakable" as the simulation progresses via slider graphical elements. More on this in a bit. In the meantime we note that the *createTurtleTypes* methods take an optional block of code as a second argument where the created turtles can be initialized immediately after being created.

The next thing for us to do is to define the simulation step. That is, to coordinate what happens with the Zombie and Human turtles as time advances. We do this in our *go* method.

²³A string is a set of characters enclosed in quotes.

²⁴In Appendix C we'll see that you can import a variety of shape types and even create your own.

²⁵This just sets the default shape of the turtle type. Shapes can be subsequently changed during the simulation as well with the `shape = "shapeName"` construct.




FIGURE 9. The shapes folder in the Package Explorer view.

```

1 def go(){
2     ask (zombies()){
3         step()
4     }
5     ask (humans()){
6         step()
7     }
8 }

```

LISTING 5. The UserObserver go method.

As we see, we've already done most of the hard work of specifying the agents' behaviors. What the observer does is *ask* the Zombie and Human turtle types to execute their *step*. This is achieved with the Logo idiom *ask*. A ReLogo entity (observer, turtle, patch, etc.) can *ask* another entity or set of entities to do things by specifying the request in a block of code after the *ask* primitive²⁶. In this case, the *zombies* and *humans* primitives²⁷ return the set of all Zombie and Human turtle types respectively. What we notice in the code

²⁶The block of code is the second argument to the *ask* primitive.

²⁷Some more examples of automatically generated primitives.

block passed to *ask* is that we don't need to specify whose *step* method should be run. It's automatically understood to be in the context of the *asked* entities.

Collecting what we've covered so far in the observer code, the UserObserver class should look like this²⁸:

²⁸A clarifying note on comments in program code. Comments in ReLogo and Groovy (and Java) are specified by `//` for single line comments and `/*comment content*/` for multiline comments. All the contents of comments are ignored in terms of program logic but they are helpful (and many say indispensable) for creating readable and manageable code.

```
1  // package declaration and imports, which we can ignore for now
2
3  class UserObserver extends BaseObserver{
4
5      /**
6       * Some comments here.
7       */
8
9      def setup(){
10         clearAll()
11         setDefaultShape(Human, "person")
12         createHumans(numHumans){
13             setxy(randomXcor(),randomYcor())
14         }
15         setDefaultShape(Zombie, "zombie")
16         createZombies(numZombies){
17             setxy(randomXcor(),randomYcor())
18             size = 2
19         }
20
21     }
22
23     def go(){
24         ask (zombies()){
25             step()
26         }
27         ask (humans()){
28             step()
29         }
30     }
31
32 }
```

LISTING 6. The UserObserver class.

1.4. Creating the graphical control and display elements. We've specified the agent (turtle) behaviors and the overall flow of our Zombies model. Here we add some controls via graphical elements. The relevant file is `UserGlobalsAndPanelFactory.groovy` (Fig. 10).

Opening this file we immediately notice some, hopefully helpful, comments on what types of elements are available. For a complete list, see Appendix B. We'll choose a regular




FIGURE 10. The UserGlobalsAndPanelFactory.groovy file in the Package Explorer view.

button with a label²⁹ for our *setup* and *go* methods. Regular buttons immediately pop back up when pressed. We'll choose a toggle button with a label for our *go* method. Toggle buttons remain pressed down when pressed until they're unpressed. They are used to execute methods repeatedly.

```
addButtonWL("setup", "Setup")
addButtonWL("go", "Go Once")
addToggleButtonWL("go", "Go")
```

We also want to create slider elements, with labels, for the *numHumans* and *numZombies* variables referenced in UserObserver (Listing 6). Sliders allow one to vary the value of a variable by dragging.

```
addSliderWL("numHumans", "Number of Humans", 1, 1, 100, 50)
addSliderWL("numZombies", "Number of Zombies", 1, 1, 10, 5)
```

²⁹A trailing WL, i.e. “with label”, is added to the graphical element methods to signify that the method takes a label as an argument. Otherwise the name of the relevant object (in the case of a button, the method referred to by the button) will be displayed.

Let's anticipate that we'd like to know, at any given time, the number of remaining Humans. We employ a monitor for this.

```
addMonitorWL("remainingHumans", "Remaining Humans", 5)
```

A monitor takes as its first argument the name (as a string) of a method in our UserObserver, which we'll have to create below. The second argument is the label of the monitor and the third argument specifies how often we'll be updating the monitor, in this case every 5 time ticks.

Our UserGlobalsAndPanelFactory should look like³⁰:

```
1 // package declaration and imports, which we can ignore for now
2
3 public class UserGlobalsAndPanelFactory
4 extends AbstractReLogoGlobalsAndPanelFactory{
5
6     public void addGlobalsAndPanelComponents(){
7
8         /**
9          * Example comments
10         */
11
12         addButtonWL("setup", "Setup")
13         addButtonWL("go", "Go Once")
14         addToggleButtonWL("go", "Go")
15         addSliderWL("numHumans", "Number of Humans", 1, 1, 100, 50)
16         addSliderWL("numZombies", "Number of Zombies", 1, 1, 10, 5)
17         addMonitorWL("remainingHumans", "Remaining Humans", 5)
18     }
19
20 }
```

LISTING 7. The UserGlobalsAndPanelFactory class.

The *remainingHumans* method to be added to the UserObserver class is as follows ³¹:

³⁰Note that all the graphical elements are specified *within* the curly braces after *addGlobalsAndPanelComponents()*, in this case on lines 7 - 17.

³¹As a note, in Groovy the use of the **return** keyword to return a value from a method is optional when the returned value is given by the last statement in the method. This is because the value of the last statement in a method is automatically returned to the method's caller. Thus we can write **count(humans())** instead of **return count(humans())** in Listing 8.

```
1 def remainingHumans(){  
2     count(humans())  
3 }
```

LISTING 8. The *remainingHumans* method in UserObserver.

The UserObserver with the additional *remainingHumans* method should now look like this:

```
1  // package declaration and imports, which we can ignore for now
2
3  class UserObserver extends BaseObserver{
4
5      /**
6       * Some comments here.
7       */
8
9      def setup(){
10         clearAll()
11         setDefaultShape(Human, "person")
12         createHumans(numHumans){
13             setxy(randomXcor(),randomYcor())
14         }
15         setDefaultShape(Zombie, "zombie")
16         createZombies(numZombies){
17             setxy(randomXcor(),randomYcor())
18             size = 2
19         }
20
21     }
22
23     def go(){
24         ask (zombies()){
25             step()
26         }
27         ask (humans()){
28             step()
29         }
30     }
31
32     def remainingHumans(){
33         count(humans())
34     }
35
36
37 }
```

LISTING 9. The UserObserver class.

1.5. Running the Zombies model. Our next step is to see what happens when we actually run the Zombies model that we've built so far. We do this by clicking on the small




FIGURE 11. The downward triangle to reveal the Run options pull down menu.




FIGURE 12. The “Zombies Model” entry which launches the Zombies model.

downward triangle to the right of the green “play” button in the toolbar at the top of the ReLogo workspace (Fig. 11). This reveals a pull down menu and we select the “Zombies Model” choice (Fig. 12).




FIGURE 13. The Repast Simphony runtime.

This launches the Repast Simphony runtime (Fig. 13). The next step is to press the “Initialize” button to initialize the runtime (Fig. 14) which loads our Zombie model (Fig. 15).

The User Panel on the left reveals the graphical elements that we created (Fig. 16). Pressing the Setup button should run the *setup* method we defined earlier in the UserObserver (Listing 9) which results in Figure 17.

At this point we can choose to repeatedly press the Go Once button and observe how the model evolves (Fig. 18) or, alternatively, we can press the Go button which will automatically advance the simulation until we un-press it. Whenever we want to reset the model, we simply press the Setup button again.




FIGURE 14. The Initialize button in the Repast Simphony runtime.




FIGURE 15. The loaded Zombie model.




FIGURE 16. The graphical elements in the User Panel.




FIGURE 17. The Zombies model after pressing the Setup button.



FIGURE 18. A closeup of the Zombies model.




FIGURE 19. The New Link icon.

1.6. Adding links to the Zombies model. In this section we'll add to the Zombies model by introducing links³². We'll create Infection link types and use them to track which Human agents have been infected by which Zombie agent. We do this, in a similar manner to how we created our Human and Zombie turtle types, by selecting the "zombies.relogo" package and then clicking on the New Link icon (Fig. 19) in the toolbar. This brings up the New Link Wizard which allows us to specify the name of our link type. We fill in the Name field with "Infection" and press the Finish button.

We slightly modify the Zombie *step* method to look like:

³²Links were introduced to Logo by NetLogo. They are edges of networks where the vertices are the turtles.

```

1  def step(){
2      def winner = maxOneOf(neighbors()){
3          count(humansOn(it))
4      }
5      face(winner)
6      forward(0.5)
7      if (count(humansHere()) > 0){
8          label = "Brains!"
9          def infectee = oneOf(humansHere())
10         infect(infectee)
11         createInfectionTo(infectee)
12     }
13     else {
14         label = ""
15     }
16 }

```

LISTING 10. New Zombie step method with lines 9 - 11 modified from the previous Zombie step (cf. line 15 in Listing 3).

We've made it so that if the Zombie finds a Human, they not only infect the Human but also create an Infection link to it (lines 9 - 11 in Listing 10). When the infectee dies any links it has are automatically removed so the newly hatched Zombie won't retain those³³. In an analogous manner to how turtle type specific methods were generated when we defined new turtle types, when we create new link types we gain access to a number of new link specific methods. *createInfectionTo* is such a method. To see a full list of the methods generated see Table 2 in Appendix A.

We'd also like to be able to play with the length of the gestation period, i.e., the period it takes for a Human to die and become a Zombie, to be able to see a network of infections forming before the infected Human dies. We add a *gestationPeriod* variable as a slider (Listing 11) and modify the Human turtle type (line 17 in Listing 12) to accomplish this.

³³It is left to the interested reader to figure out how to make these Infection links persist. There are many possible ways to accomplish this but one is to use the primitives *inInfectionNeighbors* and *createInfectionsFrom*.

```
1  // package declaration and imports, which we can ignore for now
2
3  public class UserGlobalsAndPanelFactory
4  extends AbstractReLogoGlobalsAndPanelFactory{
5
6      public void addGlobalsAndPanelComponents(){
7
8          /**
9           * Example comments
10          */
11
12          addButtonWL("setup","Setup")
13          addButtonWL("go","Go Once")
14          addToggleButtonWL("go","Go")
15          addSliderWL("numHumans", "Number of Humans", 1, 1, 100, 50)
16          addSliderWL("numZombies", "Number of Zombies", 1, 1, 10, 5)
17          addSliderWL("gestationPeriod", "Gestation", 5, 1, 30, 5)
18          addMonitorWL("remainingHumans", "Remaining Humans", 5)
19      }
20
21 }
```

LISTING 11. The UserGlobalsAndPanelFactory class with a gestationPeriod variable as a slider.

```
1  // package declaration and imports, which we can ignore for now
2
3  class Human extends BaseTurtle {
4
5      def infected = false
6      def infectionTime = 0
7
8      def step(){
9          def winner = minOneOf(neighbors()){
10              count(zombiesOn(it))
11          }
12          face(winner)
13          forward(1.5)
14
15          if (infected){
16              infectionTime++
17              if (infectionTime >= gestationPeriod){
18                  hatchZombies(1){
19                      size = 2
20                  }
21                  die()
22              }
23          }
24      }
25
26  }
```

LISTING 12. The Human turtle type modified to use a variable gestation period.

Now when we run our model, we can see larger networks of Infection links forming as we increase the gestation period (Fig. 20).




FIGURE 20. The Zombie model showing infection networks.

1.7. Data sets, outputters and external plugins. Now that we have the model running with turtles, patches and links, we'll explore the creation of simple data sets to create file outputters and even to take advantage of the many plugins available which connect to external tools. We return to our `UserObserver` class and add the following property and methods:

```
def relogoRun = 0

def remainingZombies(){
  count(zombies())
}

def timestamp(){
  ticks()
}
```

LISTING 13. Additional property and methods in the `UserObserver` class.

The *relogoRun* property will be used to differentiate between runs each time we press the setup button. The *remainingZombies* method is self explanatory. The *timestamp* method returns the current value of the ReLogo counter. This counter is useful for tracking time evolution and a call to *clearAll* re-initializes it. We'll see how to increment this timer below.

In addition, we modify the `UserObserver`'s *setup* and *go* methods in the following ways:

```
1  def setup(){
2      relogoRun++
3      clearAll()
4      setDefaultShape(Human,"person")
5      createHumans(numHumans){
6          setxy(randomXcor(),randomYcor())
7      }
8      setDefaultShape(Zombie,"zombie")
9      createZombies(numZombies){
10         setxy(randomXcor(),randomYcor())
11         size = 2
12     }
13 }
14
15 def go(){
16     ask (zombies()){
17         step()
18     }
19     ask (humans()){
20         step()
21     }
22     tick()
23 }
```

LISTING 14. Modified setup and go methods in the UserObserver class.

We're incrementing the *relogoRun* property so that we'll be able to differentiate between runs each time *setup* is executed. The *tick* primitive at the end of the *go* method increments the ReLogo counter every time *go* is executed.

The resulting full UserObserver class should now look like:


```
1 class UserObserver extends BaseObserver{
2     def relogoRun = 0
3
4     def setup(){
5         relogoRun++
6         clearAll()
7         setDefaultShape(Human,"person")
8         createHumans(numHumans){
9             setxy(randomXcor(),randomYcor())
10        }
11        setDefaultShape(Zombie,"zombie")
12        createZombies(numZombies){
13            setxy(randomXcor(),randomYcor())
14            size = 2
15        }
16    }
17
18    def go(){
19        ask (zombies()){
20            step()
21        }
22        ask (humans()){
23            step()
24        }
25        tick()
26    }
27
28    def remainingHumans(){
29        count(humans())
30    }
31
32    def remainingZombies(){
33        count(zombies())
34    }
35
36    def timestamp(){
37        ticks()
38    }
39
40 }
```

LISTING 15. The full UserObserver class after data set related modifications.




FIGURE 21. The Data Set label in the Scenario Tree panel.

The next step is to launch the Zombies model. Once it is launched, we go to the Scenario Tree panel underneath the User Panel and carry out the following steps:

- (1) Right click on the Data Sets label (Fig. 21) and choose Add DataSet.
- (2) Choose the UserObserver as the Agent Class and click on Next.
- (3) Click Add to add a Simple Agent Mapping. Left click on *current tick: double* in the source column in the second row and choose *timestamp()* from the drop down list.
- (4) Click Add again. The tick mapping should be added again. Choose *getRelogoRun*³⁴ from the drop down list.
- (5) Repeat Step 4, but choose *remainingHumans* from the list.
- (6) Repeat Step 4, but choose *remainingZombies* from the drop down list.

At this point you should see the Data Set Editor window looking like Fig. 22. Click on Next, then on Finish. Don't forget to save this setting by clicking on the save icon in the top left hand side of the Repast Symphony runtime (Fig. 23). Alternatively select Save under the File menu.

Next we'll create a file outputter. To do this we carry out the following steps:

- (1) Right click on Outputters label in the Scenario Tree (you may have to scroll down a bit).

³⁴Groovy automatically generates "getters" and "setters" for properties you define. If the property is "relogoRun" like in this example, the methods *getRelogoRun* and *setRelogoRun* are generated. This capitalization scheme, where the beginning of multiple contiguous words are capitalized, is a standard in many programming languages, including Groovy, and is referred to as "camel-casing".




FIGURE 22. The Data Set Editor window.




FIGURE 23. The Save icon in the Repast Symphony runtime toolbar.




FIGURE 24. The outputted data in the Zombies model.

- (2) Click on *timestamp* in File Columns and click the green right arrow.
- (3) Repeat for the remaining items in the column.
- (4) Click Next, then Finish (the default location for the file outputter is good for the purpose of this guide).

We've chosen the data items from our data set that we wish to output, in this case in comma-delimited form. As we did for the data set, don't forget to save this setting (Fig. 23).

Now we initialize the runtime (Fig. 14) and run the model via the setup and go buttons. At this point if we quit out of the runtime and return to our workspace we may notice that nothing has changed. However if we select our Zombies project and "Refresh" it³⁵, either by right clicking on the project and choosing Refresh from the drop down menu or by pressing the F5 button, we should see the newly generated data file (Fig. 24). Double-clicking on this file will reveal a comma-delimited four column data set.

We can also directly connect to some external tools. To demonstrate this, we launch the Zombies model again. After initializing the runtime, and running the model, we turn our attention to the External Tools buttons at the top of the runtime window (Fig. 25).

³⁵The Refresh is necessary when outside processes, in this case our running model, change files from outside the workspace.




FIGURE 25. The External Tools buttons in the Repast Simphony runtime.




FIGURE 26. The Spreadsheet wizard with the File Outputter selected.

For now we'll demonstrate how we can easily export our data into Microsoft Excel. To do this we choose the Spreadsheets plugin, the button with the calculator icon. If you have Excel on your machine, chances are the default location is correct. Otherwise select the appropriate location via the Browse button. Click Next and you should see that the File Outputter is selected (Fig. 26). Click Finish and Excel should launch with the data displayed in a spreadsheet. We recommend experimenting with the various other external tool plugins on your own.

2. A LITTLE MORE RELOGO.

We've gone through many of the basics for building models in ReLogo. While there are many more details that are beyond the scope of this getting started guide, below we include a few important items that we haven't covered yet.

2.1. @Diffusible variables in patches. Patches can have variables with the @Diffusible annotation:

```
@Diffusible
def aPatchVariable
```

Patch variables annotated this way can be manipulated over all the patches simultaneously with the observer primitives:

- diffuse(String, double)
- diffuse4(String, double)
- diffusibleAdd(String, Number)
- diffusibleApply(String, DoubleFunction)
- diffusibleDivide(String, Number)
- diffusibleMultiply(String, Number)
- diffusibleSubtract(String, Number)

2.2. @Plural annotation for turtles and links. The default plural form for turtle and link types is an 's' appended to the class name. Turtle and link types with unusual plural forms can have @Plural annotations specifying custom pluralizations.

```
@Plural("Mice")
class Mouse extends BaseTurtle {
  ...
}
```

This affects the names of the generated methods in Table 1 and Table 2. The custom plural forms replace the simple plural forms.

2.3. @Undirected/@Directed annotations for links. By default link types are directed. To specify that a link type is undirected, or even to be explicit about a link type being directed, we can use the @Undirected and @Directed annotations on a link class.




FIGURE 27. The Package Explorer’s view options down arrow.

```
@Undirected
class Infection extends BaseLink {
...
}
```

2.4. ReLogo Resource Filter. The default ReLogo workspace’s Package Explorer has a ReLogo Resource Filter selected. This hides many of the non-immediately-essential elements in a user project. At some point, however, it will be necessary to deselect this filter to access some of the hidden resources. To do this, select the Package Explorer’s view options down arrow (Fig. 27) and deselect the ReLogo Resource Filter (Fig. 28)³⁶. This will reveal the previously hidden resources. The same procedure applies for reselecting the filter as well.

2.5. ReLogo world dimensions. After disabling the ReLogo Resource Filter (Section 2.4), one of the elements that can be accessed is the parameters.xml file within the Zombies.rs

³⁶If the ReLogo Resource Filter is not visible in the drop down menu, go to the Filters... menu option and there will be an option to deselect it there.





FIGURE 28. The drop down menu for the Package Explorer.

folder. This is where the world dimensions are stored³⁷. Double-clicking the file will display the contents in an XML editor (Fig. 29) where they can be modified. Note that the dimensions are integers and “minPxcor” and “minPycor” should be less than or equal to 0.

³⁷Other parameters can be defined here as well. See the Repast Symphony documentation for details on creating and using custom parameters in your model.



Node	Content
parameters.xml	version="1.0" encoding="UTF-8"
parameters	
parameter	
parameter	
name	default_observer_minPxcor
displayName	default_observer_minPxcor
type	int
defaultValue	-16
parameter	
name	default_observer_maxPxcor
displayName	default_observer_maxPxcor
type	int
defaultValue	16
parameter	
name	default_observer_minPycor
displayName	default_observer_minPycor
type	int
defaultValue	-16
parameter	
name	default_observer_maxPycor
displayName	default_observer_maxPycor
type	int
defaultValue	16

Design Source

FIGURE 29. The parameters.xml file with ReLogo world dimensions highlighted.

2.6. Autobuilding. A ReLogo model is built when it is launched so by default autobuilding is disabled when a new ReLogo project is created. However, in the case that your workspace has autobuilding enabled and a ReLogo project is imported into the workspace, we recommend that you disable autobuilding yourself, as it won't be disabled automatically. To do this, go to the Project menu and deselect Build Automatically.

2.7. Default/Recommended Workspace Settings for ReLogo. Repast Symphony is distributed with a default workspace. If you have modified your workspace or need to use a custom workspace, the following steps will generate the optimal default workspace settings for ReLogo.

- Go to the Preferences menu option (e.g., under the Eclipse menu on Mac OS X, under the Window menu on Windows)
 - Under General → Editors → Text Editors:
 - * Check: Show line numbers
 - Under Groovy:
 - * Check: Use monospace font for JUnit
 - * Uncheck: Underline statically unknown references
 - * Uncheck: Do not use parens around methods with arguments
 - * Uncheck: Use brackets for closure arguments
 - Under Run/Debug → Launching:
 - * Check: Always launch the previously launched application
- Select the ReLogo Resource Filter (see 2.4):
- Disable Auto Building (see 2.6)

3. PARAMETER SWEEPS AND MODEL DISTRIBUTION IN REPAST SIMPHONY

3.1. Stochastic Execution and Parameter Sweeps. Most Repast models use random draws and are therefore stochastic simulations. Stochastic simulations will produce different outcomes for different random number streams, which are generally driven by choosing different random seeds. Such simulations should be executed repeatedly to explore the space of possible outcomes. Even without randomness, model sensitivity analysis parameter sweeps should be run to determine the response of the model to changes in input values. Repast provides both local and distributed tools for automatically completing stochastic execution runs and parameter sweeps. Please see the Repast Parameter Sweep Guide for more information.

3.2. Model Distribution. Repast models can be distributed to model users via the installation builder. This feature packs up your model and all of the software you need to run it, except for a properly configured Java Runtime Environment, into a single Java archive ("JAR") file that can be given to model users. The resulting installer can be executed on any system with Java version 1.6 or later; JOGL; and Java3D installed³⁸. Users simply copy the installer file onto their Windows, Mac OS, or Linux computers and then start the installer by double clicking on the file. Once the installer is started it will show an installation wizard that will prompt the user for the information needed to install the model. If desired, the installer can also be run in a command line mode.

Building an installer for a model is straightforward. Simply choose the "Build Installer for <Your Model Name Here> Model" and provide a location and name for the installer file. The installer file's default name is "setup.jar," which is suitable for most purposes. The install builder will then package and compress your model and the supporting Repast software. The resulting installer files are about 70 MB plus the size of the model code and data. 75 MB to 80 MB is a common total size.

The Repast install builder uses the IzPack system (<http://izpack.org/>). More information on installer customization and use, including command line activation, can be found on the IzPack web site.

³⁸Users can obtain free JOGL and Java3D files from the Repast website downloads page here.

APPENDIX A. GENERATED RELOGO PRIMITIVES

When a new turtle type is defined a number of methods become available to turtles, patches, links and observers. The specific methods generated when a Zombie turtle type is defined are listed in Table 1. These are similar in functionality to existing ReLogo primitives with similar names but specialized for the particular turtle type.

TABLE 1. Methods generated when a Zombie turtle type is defined.

Generated for each			
turtle	patch	link	observer
hatchZombies	sproutZombies		
zombiesHere	zombiesHere		
zombiesAt	zombiesAt		
zombiesOn	zombiesOn	zombiesOn	zombiesOn
isZombieQ	isZombieQ	isZombieQ	isZombieQ
zombies	zombies	zombies	zombies
zombie	zombie	zombie	zombie
			createZombies
			createOrderedZombies

When a new link type is defined methods are generated and become available to turtles, patches, links and observers. The specific methods created when a Connection link type are listed in Table 2. Note that there are differences depending on whether Connection is a directed or undirected link type. Again, these are similar in functionality to existing ReLogo primitives with similar names but specialized for the particular link type.

TABLE 2. Methods generated when a Connection link type is defined.

turtle	Generated for each ^{a b}		
	patch	link	observer
isConnectionQ	isConnectionQ	isConnectionQ	isConnectionQ
connections	connections	connections	connections
connection	connection	connection	connection
createConnectionFrom ^a			
createConnectionsFrom ^a			
createConnectionTo ^a			
createConnectionsTo ^a			
createConnectionWith ^b			
createConnectionsWith ^b			
inConnectionNeighborQ ^a			
inConnectionNeighbors ^a			
inConnectionFrom ^a			
myInConnections ^a			
myOutConnections ^a			
outConnectionNeighborQ ^a			
outConnectionNeighbors ^a			
outConnectionTo ^a			
connectionNeighborQ			
connectionNeighbors			
connectionWith ^b			
myConnections			

^aFor a directed link Connection.^bFor an undirected link Connection.

APPENDIX B. AVAILABLE GRAPHICAL ELEMENTS

Table 3 lists the various elements available for use in the `UserGlobalsAndPanelFactory`'s *addGlobalsAndPanelComponents* method.

TABLE 3. Elements available for use in the `UserGlobalsAndPanelFactory`'s *addGlobalsAndPanelComponents* method

Element Types	Commands ^a
Button	<code>addButton(WL)</code> ^b
State Change Button ^c	<code>addStateChangeButton(WL)</code> ^b
ToggleButton	<code>addToggleButton(WL)</code> ^b
Slider	<code>addSlider(WL)</code>
Chooser	<code>addChooser(WL)</code>
Switch	<code>addSwitch(WL)</code>
Input	<code>addInput</code>
Monitor	<code>addMonitor</code>
Global	<code>addGlobal</code>

^aThe (WL) refers to variants *with labels*.

^bInclude variants with observer specified by id.

^cState change buttons do not advance the simulation schedule.

Additionally, *Observers* are provided with a *registerModelParameterListener(String, Closure)* method which allows for registering a closure to execute when a specified model parameter (such as those defined by the Global, Slider, Chooser, Switch and Input elements) are modified. This method would typically be called from within a *setup*-like *Observer* method.

Groovy's `SwingBuilder` can also be used to create more advanced GUIs. The elements available are listed in Table 4. Listing 16 shows a code snippet example. Note that the usage pattern of the elements in Table 4 is to assign the elements to a variable and use `SwingBuilder`'s *widget* method to incorporate them into panels you create with *addPanel*³⁹.

³⁹The *addPanel* method passes its closure argument to a `SwingBuilder panel` element.

TABLE 4. Elements available for use with Groovy's SwingBuilder in the UserGlobalsAndPanelFactory's *addGlobalsAndPanelComponents* method

Element Types	Commands ^a
Panel	addPanel
Button	button(WL) ^b
ToggleButton	toggleButton(WL) ^b
Slider	slider(WL)
Chooser	chooser(WL)
Switch	rSwitch(WL)
Input	input
Monitor	monitor

^aThe (WL) refers to variants *with labels*.

^bInclude variants with observer specified by id.

```

1 def setupB = buttonWL("setup","Set up the Sim")
2 def goTB = toggleButtonWL("go","Continuous Go")
3 addPanel{
4     GridLayout(columns:1, rows:0)
5     widget(setupB)
6     widget(goTB)
7 }
```

LISTING 16. Using Groovy's SwingBuilder for building GUIs.

APPENDIX C. MORE ON TURTLE SHAPES

Any image file that's placed in the shapes folder of a ReLogo project can be referred to by name (minus suffix) within the model. Different platforms will have some minor differences in the types of image files that are readable but most of the standard ones can be used (e.g., "jpeg", "pbm", "bmp", "jpg", "wbmp", "ppm", "png", "jp2", "pgm", "gif"). In addition to regular image files, ReLogo allows one to create svg files that can either be used as images, which we refer to as "complex" svg files, or as shapes which have specified regions with fixed colors and regions whose colors can be changed within a ReLogo simulation. The default svg files that come with any newly created ReLogo project are of the latter type, which we refer to as "simple" svg files.

We recommend the free and open source SVG editor Inkscape for creating svg images. There are some keywords that can be used to define the special ReLogo properties of an svg file you create. These are specified as a comma delimited list in the "Keywords" field of the file's Document Metadata (File → Document Metadata...):

- `simple`: indicates that the svg file is simple (assumed complex if not specified)
- `rotate`: indicates whether the icon should rotate according to the turtle's heading
- `offset value`: indicates the clockwise offset (in degrees) of the shape

To fix the color of an element within a simple svg file right click on the element and select "Object Properties." In the Description field enter "fixedColor" and the element will retain its color. All other elements in a simple svg will take on the color assigned to the turtle within a simulation.